
Creating
Integrated

Applications with
Automation

Objectives

• See examples of object models from both Microsoft and third-party
applications

• Use automation to design solutions to business problems

VBA Hosts and Object Models

In this course you’ve seen many examples of working with the object model of
a VBA host like Microsoft Excel or Microsoft Word.  Remember, though,
Automation solutions are not limited to products that include VBA as their
programming language.  Only the controller in an Automation conversation
needs to have any programming language at all; the server only needs to
expose an object model.

ADO provided us with an example of an object model that is not part of a
VBA host.  The object model concepts provide the ability to extend
applications beyond the scope of the VBA host and create business solutions
that appear seamless to the end user.

The examples included here are simplified versions of real business solutions
and the possibilities for creating solutions are limited only by your and your
user’s imagination.

Letter Generator from Access

Problem

A company maintains a large customer database in Microsoft Access and
generates a number of letters, or other documents, from time to time for any
given customer. The users also want to be able to edit the stored documents
and create new ones when necessary.  Each time a user needs to send a letter,
he or she will select the recipient and the letter.  The application should then
display the letter with the recipient’s name and address filled in and allow the
user to edit and print.

Solution

We haven’t focused on Microsoft Access in this course because it doesn’t
include the Visual Basic Editor or User Forms.  Access has had its own version
of forms and its own editor since the first release.  But it is a VBA Host and
has an object model to allow not only data access, but also control of other
objects such as forms and reports.

This solution involves two applications and several different objects.  The user
selects a customer and a letter from Microsoft Access.  The selected letter is
opened in Microsoft Word and the name and address of the selected customer
is filled in.

Microsoft Access Bound Form

A standard Access form bound to the table tblCustomer is shown in Figure 1.
The user scrolls to the selected customer and clicks the “Send a Letter”
command button.

See frmCustomer
and
fdlgSendALetter in
VBA.mdb and Zoo
Gnus.Doc

Figure 1. frmCustomer in Microsoft Access is bound to tblCustomer.

The Click event of the command button calls the following event procedure:

Private Sub cmdSendALetter_Click()

  DoCmd.OpenForm "fdlgSendALetter", _

     windowmode:=acDialog

End Sub

DoCmd.OpenForm is analogous to the form.Show syntax used with User
Forms.

Letters to be Sent

Our problem states that the user must be allowed to create and change letters at
will.  As Word documents, these letters come with some very specific
attributes.  Each letter has bookmarks that are part of a standard set we’ve
given the user:  FullName, FirstName, LastName, Address, and CSZ.  The user
creates the necessary letters and adds the specific bookmarks wherever they
are needed within the document.  This is one place where a Word Template
can be very useful.

A letter might look the one shown in Figure 2:

Figure 2. Sample letter.

To add a Bookmark:

1. Select the text to be bookmarked or move the cursor to a specific insertion
point.

2. Choose Insert|Bookmark from the menu.

3. Type the name of the new bookmark.

4. Click Add.

The bookmarks dialog box for the sample document is shown in Figure 3.

Figure 3. Bookmarks for Zoo Gnus.doc

Microsoft Access Dialog Box

For this to work the user must select a letter. In this example, all the letters
have been stored in a certain directory.

The dialog box shown in Figure 4 lists all the documents in the directory and
allows the user to select one.

Figure 4. fdlgSendALetter lists all the documents in c:\My Documents\Standard
Letters.

The list of files is regenerated each time the dialog box is loaded using the
Load event.  To get the list, we used an Office 97 object discussed in the next
section, FileSearch.

The FileSearch Object

The FileSearch object allows you to search for a file on your system from
VBA. If you’ve ever tried to do this, then you’ll realize how useful the
FileSearch object can be.

The FileSearch object contains one object, the FoundFiles object. The
FoundFiles object contains any file names that are returned by the file search.

NOTE The FileSearch object is available from non-Office applications,
but only through Automation. That is, to use it from an application
such as Microsoft Visual Basic, you must first create an instance
of an Office application, such as Wordand then use the FileSearch
object model through the Word object model. Also, the FileSearch
object is available only on computers with Office 97 installed.

FileSearch Methods and Properties

The following table shows some of the methods and properties of the
FileSearch object.

Method/Property Description

NewSearch method You use this method to begin a new search.

LookIn property A string that specifies the directory to be searched.

FileType property The type of file you want to find. For example, you
can search for all files, or only for Microsoft Excel
workbooks, or Word documents, and so on.

SearchSubFolders property Specifies whether or not to search in subfolders.

FileName property The name of the file to search for. Can contain
wildcard characters.

Execute method Executes the search.

Creating the List

The RowSource property of a list box in Microsoft Access can be a table, a
query, a callback function, or a semi-colon delimited Value List.  Here we use
a value list which is built during the Load event.  The Load event procedure
for fdlgSendALetter looks like this:

Private Sub Form_Load()

    Dim varItem As Variant

    Dim ctlWordDoc As Control

    ' Initialize a variable for the listbox

    Set ctlWordDoc = Me.lstWordDocument

    ctlWordDoc.RowSource = ""

    'Use the FileSearch object to get a list of

    '  documents in a certain directory.

    With Application.FileSearch

        .FileName = "*.doc"

        ' Edit this line for your own setup, or

        '    drive the location from a table so

        '    the user can set it.

        .LookIn = "c:\My Documents\Standard Letters"

        .SearchSubFolders = False

        .Execute

        ' Fill the listbox with files that were found

        For Each varItem In .FoundFiles

            ctlWordDoc.RowSource = _

  ctlWordDoc.RowSource & varItem & ";"

        Next varItem

    End With

End Sub

Filling in the Letter

Once the user has made a selection, he or she clicks the OK button. The code
then opens the selected document in Word and replaces the text in the pre-
defined bookmark set with the values currently on the frmCustomer.

Because the Word application and document will be used in several procedures
within this form, the object variables are declared in the Declarations section
of the form, right below the Option Explicit statement as shown in Figure 5:

Figure 5. Module-level declaration of the Word object variables.

NOTE Remember that to use the Word objects within Microsoft Access,
you must set a Reference to it first.

This is the code on the Click event of the OK button, cmdOK:

Private Sub cmdOK_Click()

    Dim F As Form

    Dim retval As Boolean

    If IsNull(Me.lstWordDocument) Then

        MsgBox "Select a letter."

        Exit Sub

    End If

    ' Set a reference to the form where the data

    '    is located.

    Set F = Forms!frmCustomer

    ' Set reference to the Word Application and

    '    open the document the user wants to work with.

    Set wrdApp = New Word.Application

    wrdApp.Documents.Open (Me.lstWordDocument)

    Set wrdDoc = wrdApp.Documents(Me.lstWordDocument)

    ' Call our own function to fill in the bookmark

    retval = FillInBookmark("FullName", _

                F.FirstName & " " & F.LastName)

    retval = FillInBookmark("Address", _

                F.Address)

    retval = FillInBookmark("CSZ", _

                F.City & ", " & F.State & _

                "  " & F.ZipCode)

    retval = FillInBookmark("FirstName", _

                F.FirstName)

    ' Let the user work with the document

    wrdApp.Visible = True

    wrdDoc.Activate

End Sub

This code calls a function we wrote, FillInBookmark to find a specified
bookmark and replace the text in it with the value we send:

Private Function FillInBookmark(strBookmark As String,

varValue As Variant)

    ' First check to make sure bookmark exists

    If IsBookmark(strBookmark) Then

        ' if it does, then select it and

  '    fill in the value sent

        wrdDoc.Bookmarks(strBookmark).Text = varValue

    End If

End Function

IsBookmark is also a user-defined function:

Function IsBookmark(strBookmark As String) As Boolean

    Dim bkm As Word.Bookmark

    ' Loop through the Bookmarks collection

    '    and check names until a matching

    '    one is found.  If one is found,

    '    the function returns True.

    '    If none is found, the function

    '    returns False.

    For Each bkm In wrdDoc.Bookmarks

        If bkm.Name = strBookmark Then

            IsBookmark = True

            Exit Function

        End If

    Next bkm

End Function

The above function guards against error messages if the user doesn’t put in all
allowable bookmarks.

Letter Generator from Word – with
Microsoft Access Data

Problem

It’s started to happen – the people who use Microsoft Access are bragging to
others about how well their letter generator works.  They just put together the
document, add the appropriate bookmarks, and they’re set.  Now other people,
who don’t use Microsoft Access, want to have the same ability to pop a
customer’s address into their document.

Solution

While the code for this solution is written in Word, and not much of the code
that does the real work changes, the user interface is completely different.  To
make it easier for users to create new letters, we turned the Zoo Gnus
document from the previous example into a Word template.  The template’s
project is shown in Figure 6.

Figure 6. InfoToGo Standard Letter Project

By adding one procedure, a user form is displayed to allow the user to select a
customer.

See InfoToGo
Standard Letter.dot

Displaying the Customer Selection Dialog Box

The procedure in basSelectCustomer looks like this:

Sub DisplayForm()

    fdlgSelectCustomer.Show

End Sub

This line of code simply displays the form.  It is run from the custom toolbar
button shown in Figure 7.

Figure 7. Custom Toolbar button to select a customer.

To create a custom toolbar to run a VBA procedure follow these steps:

1. Write either a function procedure or a sub-procedure.

2. Select Tools|Customize from the menu

3. Click the New button and type in a name for the toolbar

4. Click the Commands tab and choose Macros from the Categories list.

5. Select your procedure from the Commands list and drag it to the toolbar.

6. Right-click the button on the toolbar to change properties like the name or
caption of the button.

The Select a Person button shown in Figure 7 runs the DisplayForm sub-
procedure.

The Customer Selection Dialog Box

Using the Customer Selection dialog box shown in Figure 8the user can select
a customer from the combo box and click the Fill Letter button to insert the
customer’s name and address into the selected letter.

Figure 8. The Customer Selection dialog box.

In this example, ADO is the Automation Server.  The first ADO task is to fill
the combo box with names from the table of customers stored in Microsoft
Access.  After the user selects a name and clicks the Fill Letter button, the
code must find the customer’s address information and insert it into the letter
as in the previous example.

Using ADO

In this project, a reference to ADO has been set.  Because there are several
procedures that need to make connections to an ADO resource, the connection
string is set as a constant in the Declarations section of the form module:

Option Explicit

Const conConnect = "DRIVER=Microsoft Access " & _

    "Driver (*.mdb);" & _

    "DBQ=C:\My Documents\VBA.mdb;"

If the location of the database changes, the code need only be changed in one
place.

Filling the Combo Box

Like the list box, a combo box gives the user a way to select names from a pre-
set list. Normally, the user must select a Customer ID but it’s always easier to
select a customer by name.  Therefore, this combo box has an additional
feature of providing two columns: one for the customer name and one for the
customer ID.  The user selects from the first column, Full Name,  and the

combo box, whose BoundColumn property is set to “2”,  sets the value of the
second column, Customer ID.

The combo box is then filled on the Initialize event of the form.  The code
opens an ADO Recordset and uses the AddNew and List methods of the
combo box to add each record to it:

Private Sub UserForm_Initialize()

    Dim rst As ADODB.Recordset

    Dim n As Integer

    ' Open a recordset to get the customer IDs and

    '   names for the combo box.

    Set rst = New Recordset

    rst.Open "SELECT [CustomerID], " & _

        "[LastName] & ', ' & [FirstName] As " & _

        "[FullName] " & _

        "FROM tblCustomer " & _

        "ORDER BY [LastName], [FirstName]", _

      conConnect, _

      adOpenKeyset, adLockPessimistic

    ' Loop through the recordset and add

    '    each record to the combo box.

    Do Until rst.EOF

        ' AddItem adds a record to the combo box.

        cboPerson.AddItem rst![FullName]

        ' List adds additional information to

        '   combo box.

        cboPerson.List(n, 1) = rst![CustomerID]

        cboPerson.ListIndex = n

        n = n + 1

        rst.MoveNext

    Loop

    rst.Close

End Sub

The Command Button

The user clicks the Fill Letter command button once they have made a
selection.  The code on the button’s click event uses ADO to grab only the
requested record and insert it into the letter, using the same functions coded in
the previous example:

Private Sub cmdFillLetter_Click()

    Dim rst As ADODB.Recordset

    Dim retval as Boolean

    ' Make sure the user selected something.

    If Not IsNull(cboPerson) Then

        ' Open a recordset with everything, but only

     '    for the selected customer.

        Set rst = New Recordset

        rst.Open "SELECT * FROM tblCustomer WHERE " & _

            "[CustomerID] = " & cboPerson & ";", _

        conConnect, _

        adOpenKeyset, adLockPessimistic

        retval = FillInBookmark("FullName", _

                    rst!FirstName & _

                    " " & rst!LastName)

        retval = FillInBookmark("Address", _

                    rst!Address)

        retval = FillInBookmark("CSZ", _

                    rst!City & ", " & _

                    rst!State & "  " & _

                    rst!ZipCode)

        retval = FillInBookmark("FirstName", _

                    rst!FirstName)

        rst.Close

        Unload Me

      End If

End Sub

The FillInBookmark and IsBookmark functions were cut and pasted from the
Access module and modified only slightly.  Since this code runs in Word,
explicit references to the Word object model’s hierarchy are not needed:

Function FillInBookmark(strBookmark As String, varValue

As Variant)

    ' First check to make sure bookmark exists.

    If IsBookmark(strBookmark) Then

        ' if it does, then select it and fill

  '      in the value sent.

        ActiveDocument.Bookmarks(strBookmark).Text _

 = varValue

    End If

End Function

Function IsBookmark(strBookmark As String) As Boolean

    Dim bkm As Bookmark

    For Each bkm In ActiveDocument.Bookmarks

        If bkm.Name = strBookmark Then

            IsBookmark = True

            Exit Function

        End If

    Next bkm

End Function

Outlook Text Message from Excel

Problem

Microsoft Excel, Word, and Access all provide the ability to send objects via
e-mail through their menus.  But suppose you wanted to send a plain text
message based on data in a spreadsheet.  In this example, a spreadsheet is used
to consolidate and calculate P&L information on a monthly basis.  The user
wants to send a quick bottom line summary in text format to key individuals.

For the example, we are using the Sales sheet in the VBA Excel Examples
workbook shown in Figure 9.  The monthly message, sent to a standard
distribution list, includes the name of the top salesperson for the month, his or
her dollar total, and the overall sales total.

Figure 9. Monthly Sales Information.

Solution

Microsoft Outlook is another example of a non-VBA host application that
provides a usable object model.  Prior to Outlook, sending mail messages
involving anything beyond what could be done through the menus in the
traditional Office products required using the Mail API.

Microsoft Outlook’s object model, shown in Figure 10, allows
you to access the information you see in your Microsoft

See
basSendSalesMail
in VBA.xls

Outlook folders.  In this example, we’ll use only the Application object and the
MailItem, found within the MAPIFolder.

Figure 10. Microsoft Outlook Object Model

To get started, a reference was set to the Outlook 98 Type Library in the MS
Excel spreadsheet.

SendSalesMail Procedure

This example runs from a command bar button.  When the user clicks the Send
Sales Mail button, the SendSalesMail procedure is called.  It uses information
found in the Sales worksheet to build a message and send it using Outlook.

Sub SendSalesMail()

    Dim rngStart As Range, rngData As Range

    Dim dblTotal As Double, strTop As String

    Dim dblTop As Double

    Dim otlApp As Outlook.Application

    Dim otlNewMail As Outlook.MailItem

    Dim blnResolveSuccess As Boolean

    Dim strSubject As String, strMsg As String

    ' Set subject and message

    strSubject = "Top Salesperson for " & _

        Format(DateAdd("m", -1, Date), "mmmm, yyyy")

    ActiveWorkbook.Worksheets("Sales").Activate

    ' Set active cell.

    Set rngStart = ActiveSheet.Range("A1")

    rngStart.Activate

    ' Get data region.

    Set rngData = rngStart.CurrentRegion

    ' Grab the overall sales total

    dblTotal = rngData.Cells(rngData.Rows.Count, _

rngData.Columns.Count).Value

    ' Grab name of top salesperson

    strTop = rngData.Cells(2, 1)

    ' Grab sales dollars for top salesperson

    dblTop = rngData.Cells(2, 2)

    ' Build the message

    strMsg = "The top salesperson last month was " & _

        strTop & _

        " with sales of " & _

        Format(dblTop, "Currency") & _

        "." & vbCrLf & vbCrLf & _

        "Our total sales for the month were " & _

        Format(dblTotal, "Currency") & "."

    ' Initialize Outlook objects

    Set otlApp = New Outlook.Application

    Set otlNewMail = otlApp.CreateItem(olMailItem)

    ' Create mail item

    With otlNewMail

        .Recipients.Add "The Big Boss;" & _

            "Pointy-Haired Boss Guy"

        blnResolveSuccess = .Recipients.ResolveAll

        .Subject = strSubject

        .Body = strMsg

        ' If all recipients were resolved, send.

        ' Otherwise, display to the user.

        If blnResolveSuccess Then

            .Send

        Else

            .Display

        End If

    End With

End Sub

Note the code relies on a certain format for the spreadsheet.  For example, the
top salesperson is assumed to be at the top of the list.  Code can also be written
to sort the regional values in descending order.

It is also assumed that the overall sales total is in the bottom right corner of the
region.  While a specific format is needed, the user could add columns and
rows without affecting that part of the code.

Summary

• VBA can be used to control any object model exposed to the
developer, not just objects in the host application.

• Microsoft Access is a VBA host, although its forms differ in some
ways from User Forms.

• Microsoft Word Bookmarks can be programmed to mark replaceable
text.

• Custom Toolbars are great for calling procedures.

• Microsoft Outlook’s object model provides a way for developers to
use code to send text or binary messages automatically or at the touch
of a button.


